@echo off
chcp 65001 >nul
echo 功能：创建指定数量的序列化名称的文件夹

setlocal EnableDelayedExpansion

:: 读取用户输入的数量 
set /p count=请输入要创建的文件夹数量：
if "%count%"=="" exit

:: 检查输入是否为有效数字
echo %count%| findstr /r "^[0-9][0-9]*$" >nul
if errorlevel 1 (
    echo 输入错误，请输入有效的数字！
    pause
    exit
)

:: 计算文件夹名称的补零位数
set "maxnum=%count%"
set "format=1"
:calc_format
if %maxnum% GEQ 10 (
    set /a format+=1
    set /a maxnum/=10
    goto calc_format
)

:: 创建文件夹
for /l %%i in (1,1,%count%) do (
    set "num=000%%i"
    set "folder_name=!num:~-%format%!"
    mkdir "!folder_name!"
)

exit